

4

Analysis of Example Truss by a CAS

TABLE OF CONTENTS

	Page
§4.1. Computer Algebra Systems	4-3
§4.1.1. Why Mathematica?	4-3
§4.1.2. How to Get It	4-3
§4.1.3. Programming Style and Prerequisites	4-4
§4.1.4. Class Demo Scripts	4-5
§4.2. Program Organization	4-6
§4.3. The Element Stiffness Module	4-7
§4.3.1. Module Description	4-7
§4.3.2. Programming Remarks	4-8
§4.3.3. Case Sensitivity	4-9
§4.3.4. Testing the Member Stiffness Module	4-9
§4.4. Merging a Member into the Master Stiffness	4-9
§4.5. Assembling the Master Stiffness	4-11
§4.6. Modifying the Master System	4-11
§4.7. Recovering Internal Forces	4-13
§4.8. Putting the Pieces Together	4-14
§4.8.1. The Driver Script	4-15
§4.8.2. Is All of This Worthwhile?	4-15
§4. Notes and Bibliography	4-17
§4. References	4-17
§4. Exercises	4-18

§4.1. Computer Algebra Systems

Computer algebra systems, known by the acronym CAS, are programs designed to perform symbolic and numeric manipulations following the rules of mathematics.¹ The development of such programs began in the mid 1960s. The first comprehensive system — the “granddaddy” of them all, called *Macsyma* (an acronym for Project **Mac Symbolic Manipulator**) — was developed using the programming language Lisp at MIT’s famous Artificial Intelligence Laboratory over the period 1967 to 1980.

The number and quality of symbolic-manipulation programs has expanded dramatically since the availability of graphical workstations and personal computers has encouraged interactive and experimental programming. As of this writing the leading general-purpose contenders are *Maple* and *Mathematica*.² In addition there are a dozen or so more specialized programs, some of which are available free or at very reasonable cost. See **Notes and Bibliography** at the end of the Chapter.

§4.1.1. Why Mathematica?

In the present book *Mathematica* will be used for Chapters and Exercises that develop symbolic and numerical computation for matrix structural analysis and FEM implementations. *Mathematica* is a commercial product developed by Wolfram Research, web site: <http://www.wolfram.com>. The version used to construct the code fragments presented in this Chapter is 4.1, which was commercially released in 2001. (Update: The latest version is 8, released in 2011.) The main advantages of *Mathematica* for technical computing are:

1. Availability on a wide range of platforms that range from PCs and Macs through Unix workstations. At CU Boulder, it is available through a free campus license (see §4.1).
2. Up-to-date user interface. On all machines *Mathematica* offers a graphics user interface called the Notebook front-end. This is mandatory for serious work. It provides professional typesetting for results.
3. A powerful programming language.
4. Good documentation and abundance of application books at all levels.

One common disadvantage of CAS, and *Mathematica* is no exception, is computational inefficiency in numerical calculations compared with a low-level implementation in, for instance, C or Fortran. The relative penalty can reach several orders of magnitude. For instructional use, however, the penalty is acceptable when compared to *human efficiency*. This means the ability to get FEM programs up and running in very short time, with capabilities for symbolic manipulation and graphics as a bonus.

¹ Some CAS vendors call that kind of activity “doing mathematics by computer.” It is more appropriate to regard such programs as enabling tools that help humans with complicated and error-prone manipulations. Mettle *and* metal. As of now, only humans can do mathematics.

² Another commonly used program for engineering computations: *Matlab*, does only numerical computations although a [poorly done] interface to *Maple* can be purchased as a toolbox. Historical notes: *Macsyma* died as a commercial product in 1999, although the Lisp source code of some versions is freely available. The *Maple* developer company *Waterloo Maple Inc.*, also known as *Maplesoft*, was purchased in 2009 by the Japanese software retailer Cybernet Systems. As a result, the *Matlab* symbolic toolbox is likely to be replaced.

Eligibility: students, faculty, staff and departments of all CU campuses
Platforms: Mac OSX, Windows, Linux, Solaris, AIX, HP-UX

How to Get It

1) Download and install software as per instructions at <http://oit.colorado.edu/software-hardware/site-licenses/mathematica> or request an installation CD from the Site Licensing office: sitelic@colorado.edu, 303-492-8995

2) Register your copy online at <http://register.wolfram.com/> using campus license number L2437-5121 plus your computer MathID number. Be sure to use your CU mail address.

3) A password (unique to your computer) will be forwarded to you via email from Site Licensing

ITS Support

Question & tech support problems: send email to sitelic@colorado.edu
 Periodic hands-on workshops are available for those new to the software, click on [Workshops](#) on the Web page given above.
 Details (for example: what is a MathID?) about licensing & support are posted at the above web site.

FIGURE 4.1. Instructions to get *Mathematica 8* for free from CU-OIT (Office of Information Technology).

§4.1.2. How to Get It

Starting 1 August 2007, a free one-year license from CU’s Information Technology Services (ITS) is available. Students may renew this license as long as they are registered. See Figure 4.1 for the “How to Get It” instructions.

If you plan to keep *Mathematica* for a longer time, an academic version is available. Registered students may also purchase the student version for about \$150 at the UMC bookstore.³ You will need to show proof you are a bona-fide student at the register.⁴ If you are not on campus (e.g. a CAETE student) you may purchase it directly at the vendor’s web site <http://www.wolfram.com>. Again proof of registration must be provided.

The student version is cheap, since the standard personal license costs over \$1K and company licenses go for over \$3K per seat. Unlike other commercial software products, you get the full thing; no capabilities are emasculated. But terms are strict: once installed on your laptop or desktop, it cannot be transferred to another computer since the license is forever keyed to the disk identification. (For example, it won’t launch if the disk is erased or replaced.)

§4.1.3. Programming Style and Prerequisites

The following material assumes that you are a moderately experienced user of *Mathematica*, or

³ This was written in 2001; probably this option does not exist anymore.

⁴ Check for discounts when new versions come out; unsold previous-version copies may go for as little as \$50.

are willing to learn to be one. See **Notes and Bibliography** for a brief discussion of tutorial and reference materials in case you are interested.

The best way to learn it from scratch is trying it as a calculator and using online help as needed. Practice with the program until you reach the level of writing functions, modules and scripts with relative ease. With the Notebook interface and a good primer it takes only a few hours. When approaching that level you may notice that functions in *Mathematica* display many aspects similar to C.⁵ You can exploit this similarity if you are proficient in that language. But *Mathematica* functions do have some unique aspects, such as matching arguments by pattern, and the fact that internal variables are global unless otherwise made local.⁶

Modification of function arguments should be avoided because it may be difficult to trace side effects. The programming style enforced here outlaws output arguments and a function can only return its name. But since the name can be a list of arbitrary objects the restriction is not serious.⁷

Our objective is to develop a symbolic program written in *Mathematica* that solves the example plane truss as well as some symbolic versions thereof. The program will rely heavily on the development and use of *functions* implemented using the `Module` construct of *Mathematica*. Thus the style will be one of procedural programming.⁸ The program will not be particularly modular (in the computer science sense) because *Mathematica* is not suitable for that programming style.⁹

The code presented in §4.2 through §4.8 uses a few language constructs that may be deemed as advanced, and these are briefly noted in the text so that appropriate reference to the *Mathematica* reference manual can be made.

§4.1.4. Class Demo Scripts

The cell scripts shown in Figures 4.2 and 4.3 will be used to illustrate the organization of a Notebook file and the “look and feel” of some basic *Mathematica* commands. These scripts will be demonstrated in class from a laptop.

⁵ Simple functions can be implemented in *Mathematica* directly, for instance `DotProduct[x_, y_] := x.y`; more complicated functions are handled by the `Module` construct. These constructs are called *rules* by computer scientists.

⁶ In *Mathematica* everything is a function, including programming constructs. Example: in C `for` is a loop-opening keyword, whereas in *Mathematica* `For` is a function that runs a loop according to its arguments.

⁷ Such restrictions on arguments and function returns are closer in spirit to C than *Fortran* although you can of course modify C-function arguments using pointers — exceedingly dangerous but often unavoidable.

⁸ The name `Module` should not be taken too seriously: it is far away from the concept of modules in Ada, Modula, Oberon or *Fortran 90*. But such precise levels of interface control are rarely needed in symbolic languages.

⁹ Indeed none of the CAS packages in popular use is designed for strong modularity because of historical and interactivity constraints.




FIGURE 4.2. Example cell for class demo.




FIGURE 4.3. Another example cell for class demo. (Note: displayed results were obtained with *Mathematica* version 4.2. Integration answers from versions 5 and up are quite different.)




FIGURE 4.4. Example truss program: (a) organization by function; (b) organization by cell and module names.

§4.2. Program Organization

The overall organization of the example truss program is flowcharted in Figure 4.4. Figure 4.4(a) illustrates the program as divided into functional tasks. For example the driver program calls the stiffness assembler, which in turn calls two functions: form bar stiffness and merge into master stiffness. This kind of functional division would be provided by any programming language.

On the other hand, Figure 4.4(b) is *Mathematica* specific. It displays the names of the module that implement the functional tasks and the cells where they reside. (Those program objects: modules and cells, are described in the sections below.)

The following sections describe the code segments of Figure 4.4(b), one by one. For tutorial purposes this is done in a “bottom up” fashion, that is, going cell by cell, from left to right and bottom to top.

§4.3. The Element Stiffness Module

As our first FEM code segment, the top box of Figure 4.5 shows a module that evaluates and returns the 4×4 stiffness matrix of a plane truss member (two-node bar) in global coordinates. The text in that box of that figure is supposed to be placed on a Notebook cell. Executing the cell, by clicking on it and hitting an appropriate key (<Enter> on a Mac), gives the output shown in the bottom box. The contents of the figure is described in further detail below.

§4.3.1. Module Description

The stiffness module is called `ElemStiff2DTwoNodeBar`. Such descriptive names are permitted by the language. This reduces the need for detailed comments.

The module takes two arguments:

`{{x1,y1},{x2,y2}}` A two-level list¹⁰ containing the $\{x, y\}$ coordinates of the bar end

¹⁰ A level-one list is a sequence of items enclosed in curly braces. For example: $\{x1,y1\}$ is a list of two items. A level-two list is a list of level-one lists. An important example of a level-two list is a matrix.




FIGURE 4.5. Module `ElemStiff2DTwoNodeBar` to form the element stiffness of a 2D 2-node truss element in global coordinates. Test statements (in blue) and test output.

nodes labelled as 1 and 2.¹¹

`{ Em, A }`

A one-level list containing the bar elastic modulus, E and the member cross section area, A . See §4.3.3 as to why name E cannot be used.

The use of the underscore after argument item names in the declaration of the `Module` is a requirement for pattern-matching in *Mathematica*. If, as recommended, you have learned functions and modules this language-specific feature should not come as a surprise.

The module name returns the 4×4 member stiffness matrix internally called `Ke`. The logic that leads to the formation of that matrix is straightforward and need not be explained in detail. Note, however, the elegant direct declaration of the matrix `Ke` as a level-two list, which eliminates the fiddling around with array indices typical of low-level programming languages. The specification format in fact closely matches the mathematical expression given as (2.18) in Chapter 2.

§4.3.2. Programming Remarks

The function in Figure 4.5 uses several intermediate variables with short names: `dx`, `dy`, `s`, `c` and `L`. It is strongly advisable to make these symbols *local* to avoid potential names clashes somewhere else.¹² In the `Module[...]` construct this is done by listing those names in a list immediately

¹¹ These are called the *local node numbers*, and replace the i, j of previous Chapters. This is a common FEM programming practice.

¹² The “global by default” choice is the worst one, but we must live with the rules of the language.

after the opening bracket. Local variables may be *initialized* when they are constants or simple functions of the argument items; for example on entry to the module `dx=x2-x1` initializes variable `dx` to be the difference of x node coordinates, namely $\Delta x = x_2 - x_1$.

The `Return` statement fulfills the same purpose as in C or Fortran 90. *Mathematica* guides and textbooks advise against the use of that and other C-like constructs. The writer strongly disagrees: the `Return` statement makes clear what the `Module` gives back to its invoker and is self-documenting.

§4.3.3. Case Sensitivity

Mathematica, like most recent computer languages, is case sensitive so that for instance `E` is not the same as `e`. This is fine. But the language designer decided that names of system-defined objects such as built-in functions and constants must begin with a capital letter. Consequently the liberal use of names beginning with a capital letter may run into clashes. For example you cannot use `E` because of its built-in meaning as the base of natural logarithms.¹³

In the code fragments presented throughout this book, identifiers beginning with upper case are used for objects such as stiffness matrices, modulus of elasticity, and cross section area. This follows established usage in Mechanics. When there is danger of clashing with a protected system symbol, additional lower case letters are used. For example, `Em` is used for the elastic modulus instead of `E` because (as noted above) the latter is a reserved symbol.

§4.3.4. Testing the Member Stiffness Module

Following the definition of `ElemStiff2DTwoNodeBar` in Figure 4.5 there are several statements that constitute the *module test script*. This script calls the module and prints returned results. Two cases are tested. First, the stiffness of member (3) of the example truss, using all-numerical values. Next, some of the input arguments for the same member are given symbolic names so they stand for variables. For example, the elastic module is given as `Em` instead of 100 as in the foregoing test. The print output of the test is shown in the lower portion of Figure 4.5.

The first test returns the member stiffness matrix (2.21), as may be expected. The second test returns a symbolic form in which three symbols appear: the coordinates of end node 2, which is taken to be located at $\{L, L\}$ instead of $\{10, 10\}$, `A`, which is the cross-section area and `Em`, which is the elastic modulus. Note that the returning matrix `Ke` is subject to a `Simplify` step before printing it, the reason for which is the subject of an Exercise. The ability to carry along variables is of course a fundamental capability of any CAS, and the main reason for which such programs are used.

§4.4. Merging a Member into the Master Stiffness

The next code fragment, listed in Figure 4.6, is used in the assembly step of the DSM. Module `MergeElemIntoMasterStiff` receives the 4×4 element stiffness matrix formed by `FormElemStiff2DNodeBar` and “merges” it into the master stiffness. It takes three arguments:

`Ke` The 4×4 member stiffness matrix to be merged. This is a level-two list.

¹³ In retrospect this appears to have been a highly questionable decision. System defined names should have been identified by a reserved prefix or postfix to avoid surprises, as done in *Macsyma* or *Maple*. *Mathematica* issues a warning message, however, if an attempt to redefine a “protected symbol” is made.

```

MergeElemIntoMasterStiff[Ke_,eftab_,Kin_]:=Module[
  {i,j,ii,jj,K=Kin},
    For [i=1, i<=4, i++, ii=eftab[[i]];
      For [j=i, j<=4, j++, jj=eftab[[j]];
        K[[jj,ii]]=K[[ii,jj]]+Ke[[i,j]]
      ]
    ]; Return[K]
];
K=Table[0,{6},{6}];
Print["Initialized master stiffness matrix:"];
Print[K//MatrixForm]
Ke=ElemStiff2DTwoNodeBar[{{0,0},{10,10}},{100,2*sqrt[2]};
Print["Member stiffness matrix:"]; Print[Ke//MatrixForm];
K=MergeElemIntoMasterStiff[Ke,{1,2,5,6},K];
Print["Master stiffness after member merge:"];
Print[K//MatrixForm];

Initialized master stiffness matrix:
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )

Member stiffness matrix:
( 10 10 -10 -10 )
( 10 10 -10 -10 )
( -10 -10 10 10 )
( -10 -10 10 10 )

Master stiffness after member merge:
( 10 10 0 0 -10 -10 )
( 10 10 0 0 -10 -10 )
( 0 0 0 0 0 0 )
( 0 0 0 0 0 0 )
( -10 -10 0 0 10 10 )
( -10 -10 0 0 10 10 )

```

FIGURE 4.6. Module `MergeElemIntoMasterStiff` to merge a 4×4 bar element stiffness into the master stiffness matrix. Test statements (in blue) and test output.

`eftab` The column of the Element Freedom Table or EFT, defined in §3.5.1, appropriate to the member being merged. Recall that the EFT lists the global equation numbers for the four member degrees of freedom, cf. (3.26) in Chapter 3. This is a level-one list consisting of 4 integers.

`Kinp` The incoming 6×6 master stiffness matrix. This is a level-two list.

`MergeElemIntoMasterStiff` returns, as module name, the updated master stiffness matrix internally called `K` with the member stiffness merged in. We encounter here a novelty: an input-output argument. Because a formal argument cannot be modified, the situation is handled by copying the incoming `Kinp` into `K` (a local variable declared in the list that follows the `Module` name) on entry. That is the copy updated and returned via the `Return` statement. The implementation has a strong C flavor with two nested `For` loops. Because the iterators are very simple, nested `Do` loops could have been used as well.

The statements after the module provide a simple test. Before the first call to this function, the master stiffness matrix must be initialized to a zero 6×6 array. This is done in the first test statement using the `Table` function. The test member stiffness matrix is that of member (3) of the example




FIGURE 4.7. Module `AssembleMasterStiffOfExampleTruss` that forms the 6×6 master stiffness matrix of the example truss. Test statements (in blue) and test output.

truss, and is obtained by calling `ElemStiff2DTwoNodeBar`. The EFT is $\{1, 2, 5, 6\}$ since element freedoms 1,2,3,4 map into global freedoms 1,2,5,6. Running the test statements yields the listing given in Figure 4.6. The output is as expected.

§4.5. Assembling the Master Stiffness

The module `AssembleMasterStiffOfExampleTruss`, listed in the top box of Figure 4.7, makes use of the foregoing two modules: `ElemStiff2DTwoNodeBar` and `MergeElemIntoMasterStiff`, to form the master stiffness matrix of the example truss. The initialization of the stiffness matrix array in `K` to zero is done by the `Table` function of `Mathematica`, which is handy for initializing lists. The remaining statements are self explanatory. The module is similar in style to C functions with no arguments. All the example truss data is “wired in.”

The output from the test script is shown in the lower box of Figure 4.7. The output stiffness matches that in Equation (3.12), as can be expected if all code fragments used so far work correctly.

§4.6. Modifying the Master System

Following the assembly process the master stiffness equations $\mathbf{K}\mathbf{u} = \mathbf{f}$ must be modified to account for single-freedom displacement boundary conditions. This is done through the computer-oriented equation modification process outlined in §3.5.2.

Module `ModifiedMasterStiffForDBC` and `ModifiedMasterForcesForDBC` carry out this modification for \mathbf{K} and \mathbf{f} , respectively. These two modules are listed in the top box of Figure 4.8, along with test statements. The logic of both functions is considerably simplified by assuming that *all prescribed displacements are zero*. That is, the BCs are homogeneous. (The more general case of nonzero prescribed values is treated in Part III of the book.) The module has two arguments:

- pdof A list of the prescribed degrees of freedom identified by their global number. For the example truss this list contains three entries: $\{1, 2, 4\}$.

```

ModifiedMasterStiffForDBC[pdof_,K_] := Module[
  {i,j,k,nk=Length[K],np=Length[pdof],Kmod=K},
  For [k=1,k<=np,k++, i=pdof[[k]];
    For [j=1,j<=nk,j++, Kmod[[i,j]]=Kmod[[j,i]]=0];
    Kmod[[i,i]]=1];
  Return[Kmod]
];
ModifiedMasterForcesForDBC[pdof_,f_] := Module[
  {i,k,np=Length[pdof],fmod=f},
  For [k=1,k<=np,k++, i=pdof[[k]]; fmod[[i]]=0];
  Return[fmod]
];
K=Array[Kij,{6,6}]; Print["Assembled master stiffness:"];
Print[K//MatrixForm];
K=ModifiedMasterStiffForDBC[{1,2,4},K];
Print["Master stiffness modified for displacement B.C.:"];
Print[K//MatrixForm];
f=Array[fi,{6}]; Print["Force vector:"]; Print[f];
f=ModifiedMasterForcesForDBC[{1,2,4},f];
Print["Force vector modified for displacement B.C.:"]; Print[f];

```

```

Assembled master stiffness:
( Kij[1, 1] Kij[1, 2] Kij[1, 3] Kij[1, 4] Kij[1, 5] Kij[1, 6]
  Kij[2, 1] Kij[2, 2] Kij[2, 3] Kij[2, 4] Kij[2, 5] Kij[2, 6]
  Kij[3, 1] Kij[3, 2] Kij[3, 3] Kij[3, 4] Kij[3, 5] Kij[3, 6]
  Kij[4, 1] Kij[4, 2] Kij[4, 3] Kij[4, 4] Kij[4, 5] Kij[4, 6]
  Kij[5, 1] Kij[5, 2] Kij[5, 3] Kij[5, 4] Kij[5, 5] Kij[5, 6]
  Kij[6, 1] Kij[6, 2] Kij[6, 3] Kij[6, 4] Kij[6, 5] Kij[6, 6] )
Master stiffness modified for displacement B.C.:
( 1 0 0 0 0 0
  0 1 0 0 0 0
  0 0 Kij[3, 3] 0 Kij[3, 5] Kij[3, 6]
  0 0 0 1 0 0
  0 0 Kij[5, 3] 0 Kij[5, 5] Kij[5, 6]
  0 0 Kij[6, 3] 0 Kij[6, 5] Kij[6, 6] )
Force vector:
{fi[1], fi[2], fi[3], fi[4], fi[5], fi[6]}
Force vector modified for displacement B.C.:
{0, 0, fi[3], 0, fi[5], fi[6]}

```

FIGURE 4.8. Modules `ModifiedMasterStiffForDBC` and `ModifiedMasterForceForDBC` that modify the master stiffness matrix and force vector of a truss to impose displacement BCs. Test statements (in blue) and test output.

K The master stiffness matrix **K** produced by the assembly process.

The function clears appropriate rows and columns of **K**, places ones on the diagonal, and returns the modified **K** as function value. The only slightly fancy thing in this module is the use of the *Mathematica* function `Length` to extract the number of prescribed displacement components: `Length[pdof]` here will return the value 3, which is the length of the list `pdof`. Similarly `nk=Length[K]` assigns 6 to `nk`, which is the order of matrix **K**. Although for the example truss these values are known *a priori*, the use of `Length` serves to illustrate a technique that is heavily used in more general code.

Module `ModifiedMasterForcesForDBC` has similar structure and logic and need not be described in detail. It is important to note that for homogeneous BCs the modules are independent of each other and may be called in any order. On the other hand, if there were nonzero prescribed displacements present the force modification must be done *before* the stiffness modification. This is because

```

IntForce2DTwoNodeBar[{{x1_,y1_},{x2_,y2_}},{Em_,A_},eftab_,u_]:=
Module[ {c,s,dx=x2-x1,dy=y2-y1,L,ix,iy,jx,jy,ubar,e},
  L=Sqrt[dx^2+dy^2]; c=dx/L; s=dy/L; {ix,iy,jx,jy}=eftab;
  ubar={c*u[[ix]]+s*u[[iy]],-s*u[[ix]]+c*u[[iy]],
        c*u[[jx]]+s*u[[jy]],-s*u[[jx]]+c*u[[jy]]};
  e=(ubar[[3]]-ubar[[1]])/L; Return[Em*A*e]
];
p =IntForce2DTwoNodeBar[{{0,0},{10,10}},{100,2*Sqrt[2]},
  {1,2,5,6},{0,0,0,0,0.4,-0.2}];
Print["Member int force (numerical):"]; Print[N[p]];
p =IntForce2DTwoNodeBar[{{0,0},{L,L}},{Em,A},
  {1,2,5,6},{0,0,0,0,ux3,uy3}];
Print["Member int force (symbolic):"]; Print[Simplify[p]];

```

```

Member int force (numerical):
2.82843
Member int force (symbolic):
  A Em (ux3 + uy3)
    2 L

```

FIGURE 4.9. Module IntForce2DTwoNodeBar for computing the internal force in a bar element. Test statements (in blue) and test output.

stiffness coefficients that are cleared in the latter are needed for modifying the force vector.

The test statements are purposely chosen to illustrate another feature of *Mathematica*: the use of the Array function to generate subscripted symbolic arrays of one and two dimensions. The test output is shown in the bottom box of Figure 4.8, which should be self explanatory. The force vector and its modified form are printed as row vectors to save space.

§4.7. Recovering Internal Forces

Since *Mathematica* provides built-in matrix operations for solving a linear system of equations and multiplying matrices by vectors, we do not need to write application functions for the solution of the modified stiffness equations, and for the recovery of nodal forces as $\mathbf{f} = \mathbf{K}\mathbf{u}$. Consequently, the last application functions we need are those for internal force recovery.

Function IntForce2DTwoNodeBar listed in the top box of Figure 4.9 computes the internal force in an individual bar element. It is somewhat similar in argument sequence and logic to ElemStiff2DTwoNodeBar of Figure 4.5. The first two arguments are identical. Argument eftab provides the Element Freedom Table array for the element. The last argument, u, is the vector of computed node displacements.

The logic of IntForce2DTwoNodeBar is straightforward and follows the method outlined in §3.2.1. Member joint displacements $\bar{\mathbf{u}}^{(e)}$ in local coordinates $\{\bar{x}, \bar{y}\}$ are recovered in array ubar, then the longitudinal strain $e = (\bar{u}_{xj} - \bar{u}_{xi})/L$ and the internal (axial) force $p = EAe$ is returned as function value. As coded the function contains redundant operations because entries 2 and 4 of ubar (that is, components \bar{u}_{yi} and \bar{u}_{yj}) are not actually needed to get p , but were kept to illustrate the general backtransformation of global to local displacements.

Running this function with the test statements shown after the module produces the output shown in the bottom box of Figure 4.9. The first test is for member (3) of the example truss using the actual nodal displacements (3.17). It also illustrates the use of the *Mathematica* built in function

```

IntForcesOfExampleTruss[u_]:= Module[{f=Table[0,{3]}},
  f[[1]]=IntForce2DTwoNodeBar[{{0,0},{10,0}},{100,1},{1,2,3,4},u];
  f[[2]]=IntForce2DTwoNodeBar[{{10,0},{10,10}},{100,1/2},{3,4,5,6},u];
  f[[3]]=IntForce2DTwoNodeBar[{{0,0},{10,10}},{100,2*Sqrt[2]},
    {1,2,5,6},u];
  Return[f]
];
f=IntForcesOfExampleTruss[{0,0,0,0,0.4,-0.2}];
Print["Internal member forces in example truss:"];Print[N[f]];

```

```

Internal member forces in example truss:
{0., -1., 2.82843}

```

FIGURE 4.10. Module `IntForceOfExampleTruss` that computes internal forces in the 3 members of the example truss. Test statements (in blue) and test output.

`N` to produce output in floating-point form. The second test does a symbolic calculation in which several argument values are fed in variable form.

The top box of Figure 4.10 lists a higher-level function, `IntForceOfExampleTruss`, which has a single argument: `u`. This is the complete 6-vector of joint displacements \mathbf{u} . This function calls `IntForce2DTwoNodeBar` three times, once for each member of the example truss, and returns the three member internal forces thus computed as a 3-component list.

The test statements listed after `IntForcesOfExampleTruss` feed the displacement solution (3.24) to the module. Running the test produces the output shown in the lower box of Figure 4.10. The internal forces are $F^{(1)} = 0$, $F^{(2)} = -1$ and $F^{(3)} = 2\sqrt{2} = 2.82843$, in agreement with the values found in §3.4.2.

§4.8. Putting the Pieces Together

After all this development and testing effort, documented in Figures 4.5 through 4.10, we are ready to make use of all these bits and pieces of code to analyze the example plane truss. This is actually done with the logic shown in Figure 4.11. This particular piece of code is called the *driver script*. Note that it is *not* a Module. It uses the seven previously described modules

```

ElemStiff2DTwoNodeBar
MergeElemIntoMasterStiff
AssembleMasterStiffOfExampleTruss
ModifiedMasterStiffForDBC
ModifiedMasterForcesForDBC
IntForce2DTwoNodeTruss
IntForcesOfExampleTruss

```

(4.1)

These functions must have been defined (“compiled”) at the time the driver scripts described below are run. A simple way to making sure that all of them are defined is to put all these functions in the same Notebook file and to mark them as *initialization cells*. These cells may be executed by picking up Kernel → Initialize → Execute Initialization.¹⁴ (An even simpler scheme would to group them all in one cell, but that would make placing separate test statements messy.)

¹⁴ In *Mathematica* versions 6 and higher, there is an initialization shortcut.




FIGURE 4.11. Driver script for numerical analysis of example truss and its output.

For a hierarchical version of (4.1), see the last CAETE slide.

§4.8.1. The Driver Script

The code listed in the top box of Figure 4.11 first assembles the master stiffness matrix through `AssembleMasterStiffOfExampleTruss`. Next, it applies the displacement boundary conditions through `ModifiedMasterStiffForDBC` and `ModifiedMasterForcesForDBC`. Note that the modified stiffness matrix is placed into `Kmod` rather than `K` to save the original form of the master stiffness for the reaction force recovery later. The complete displacement vector is obtained by the matrix calculation

$$\mathbf{u} = \text{Inverse}[\mathbf{Kmod}] \cdot \mathbf{fmod} \quad (4.2)$$

This statement takes advantage of two built-in *Mathematica* functions. `Inverse` returns the inverse of its matrix argument¹⁵ The dot operator signifies matrix multiply (here, matrix-vector multiply.) The enclosing `Simplify` function in Figure 4.11 is asked to simplify the expression of vector `u` in case of symbolic calculations; it is actually redundant if all computations are numerical.

The remaining statements recover the node vector including reactions via the matrix-vector multiply $\mathbf{f} = \mathbf{K} \cdot \mathbf{u}$ (recall that `K` contains the unmodified master stiffness matrix) and the member internal forces `p` through `IntForcesOfExampleTruss`. The code prints `u`, `f` and `p` as row vectors to save space.

Running the script of the top box of Figure 4.11 produces the output shown in the bottom box of that figure. The results confirm the hand calculations of Chapter 3.

§4.8.2. Is All of This Worthwhile?

At this point you may wonder whether all of this work is worth the trouble. After all, a hand calculation (typically helped by a programable calculator) would be quicker in terms of flow time. Typing and debugging the *Mathematica* fragments displayed here took the writer about six hours (although about two thirds of this was spent in editing and getting the fragment listings into the

¹⁵ This is a highly inefficient way to solve $\mathbf{Ku} = \mathbf{f}$ if the linear system becomes large. It is done here for simplicity.




FIGURE 4.12. Driver script for symbolic analysis of example truss and its output.

Chapter.) For larger problems, however, *Mathematica* would certainly beat hand-plus-calculator computations, the cross-over typically appearing for 10 to 20 equations. For up to about 500 equations and using floating-point arithmetic, *Mathematica* gives answers within minutes on a fast PC or Mac with sufficient memory but eventually runs out of steam at about 1000 equations. For a range of 1000 to about 50000 equations, *Matlab*, using built-in sparse solvers, would be the best compromise between human and computer flow time. Beyond 50000 equations a program in a low-level language, such as C or Fortran, would be most efficient in terms of computer time.¹⁶

One distinct advantage of computer algebra systems emerges when you need to *parametrize* a small problem by leaving one or more problem quantities as variables. For example suppose that the applied forces on node 3 are to be left as f_{x3} and f_{y3} . You replace the last two components of array *p* as shown in the top box of Figure 4.12, execute the cell and shortly get the symbolic answer shown in the bottom box of that figure. This is the answer to an infinite number of numerical problems. Although one may try to undertake such studies by hand, the likelihood of errors grows rapidly with the complexity of the system. Symbolic manipulation systems can amplify human abilities in this regard, as long as the algebra “does not explode” because of combinatorial complexity. Examples of such nontrivial calculations will appear throughout the following Chapters.

Remark 4.1. The “combinatorial explosion” danger of symbolic computations should be always kept in mind. For example, the numerical inversion of a $N \times N$ matrix is a $O(N^3)$ process, whereas symbolic inversion goes as $O(N!)$. For $N = 48$ the floating-point numerical inverse will be typically done in a fraction of a second. But the symbolic adjoint will have $48! = 12413915592536072670862289047373375038521486354677760000000000$ terms, or $O(10^{61})$. There may be enough electrons in our Universe to store that, but barely ...

¹⁶ The current record for FEM structural applications is about 100 million equations, done on a massively parallel super-computer (ASCI Red at SNL). Fluid mechanics problems with over 500 million equations have been solved.

Notes and Bibliography

The hefty *Mathematica Book* [420] is a reference manual. Since the contents are available online (click on **Help** in topbar) as buying the printed book makes little sense.¹⁷

There is a nice tutorial available by Glynn and Gray [163], list: \$35, dated 1999. (Theodore Gray invented the Notebook front-end that appeared in version 2.2.) It can be also purchased on CDROM from MathWare, Ltd, P. O. Box 3025, Urbana, IL 61208, e-mail: info@mathware.com. The CDROM is a hyperlinked version of the book that can be installed on the same directory as *Mathematica*. More up to date and comprehensive is the recent appeared *Mathematica Navigator* in two volumes [338,339]; list: \$69.95 but used copies are discounted, down to about \$20. The recently appeared *Mathematica Cookbook* by S. Mangano, published by O'Reilly, is being reviewed as of this writing; if useful, it will be placed in the reference list.

Beyond these, there are many books at all levels that expound on the use of *Mathematica* for various applications ranging from pure mathematics to physics and engineering. A web search on September 2003 done on www3.addall.com hit 150 book titles containing *Mathematica*, compared to 111 for *Maple* and 148 for *Matlab*. A google search (August 2005) hits 3,820,000 pages containing *Mathematica*, but here *Matlab* wins with 4,130,000.

Wolfram Research hosts the MathWorld web site at <http://mathworld.wolfram.com>, maintained by Eric W. Weisstein. It is essentially a hyperlinked, on-line version of his *Encyclopædia of Mathematics* [409].

To close the topic of symbolic versus numerical computation, here is a nice summary by A. Grozin, posted on the Usenet:

“Computer Algebra Systems (CASs) are programs [that] operate with formulas. Mathematica is a powerful CAS (though quite expensive). Other CASs are, e.g., Maple, REDUCE, MuPAD <...>. There are also quite powerful free CASs: Maxima and Axiom. In all of these systems, it is possible to do some numerical calculations (e.g., to evaluate the formula you have derived at some numerical values of all parameters). But it is a very bad idea to do large-scale numerical work in such systems: performance will suffer. In some special cases (e.g., numerical calculations with very high precision, impossible at the double-precision level), you can use Mathematica to do what you need, but there are other, faster ways to do such things.

There are a number of programs to do numerical calculations with usual double-precision numbers. One example is Matlab; there are similar free programs, e.g., Octave, Scilab, R, ... Matlab is very good and fast in doing numerical linear algebra: if you want to solve a system of 100 linear equations whose coefficients are all numbers, use Matlab; if coefficients contain letters (symbolic quantities) and you want the solution as formulas, use Mathematica or some other CAS. Matlab can do a limited amount of formula manipulations using its “symbolic toolbox,” which is an interface to a cut-down Maple. It’s a pain to use this interface: if you want Maple, just use Maple.”

References

Referenced items have been moved to Appendix R.

¹⁷ The fifth edition, covering version 5, lists for \$49.95 but older editions are heavily discounted on the web, some under \$2. There are no printed manual versions for version 6 and higher.

Homework Exercises for Chapter 4 Analysis of Example Truss by a CAS

Before doing any of these Exercises, download the *Mathematica* Notebook file `ExampleTruss.nb` from the course web site. (Go to Chapter 4 Index and click on the link). Open this Notebook file using version 4.1 or later one. The first six cells contain the modules and test statements listed in the top boxes of Figures 4.5 through 4.10. These six are marked as *initialization cells*. Before running driver scripts, they should be executed by picking up Kernel → Evaluation → Execute Initialization (or equivalent in newer versions).

Verify that the output of those six cells agrees with that shown in the bottom boxes of those figures. Then execute the driver script in Cell 7 by clicking on the cell and pressing the appropriate key: <Enter> on a Mac, <Shift-Enter> on a Windows PC. Compare the output with that shown in 4.11. If the output checks out, you may proceed to the Exercises.

EXERCISE 4.1 [C:10] Explain why the `Simplify` command in the test statements of Figure 4.5 says $L > 0$. (One way to figure this out is to just say `Ke=Simplify[Ke]` and look at the output. Related question: why does *Mathematica* refuse to simplify `Sqrt[Lexp2]` to L unless one specifies the sign of L in the `Simplify` command?

EXERCISE 4.2 [C:10] Explain the logic of the `For` loops in the merge function `MergeElemIntoMasterStiff` of Figure 4.6. What does the operator `+=` do? (If you are a C programmer, all of this should be easy.)

EXERCISE 4.3 [C:10] Explain the reason behind the use of `Length` in the modules of Figure 4.6. Why not simply set `nk` and `np` to 6 and 3, respectively?

EXERCISE 4.4 [C:15] Of the seven modules listed in Figures 4.5 through 4.10, with names collected in (4.1), two can be used only for the example truss, three can be used for any plane truss, and two can be used for other structures analyzed by the DSM. Identify which ones, and briefly state the reasons for your classification.

EXERCISE 4.5 [C:20] Modify the modules `AssembleMasterStiffOfExampleTruss`, `IntForcesOfExampleTruss`, and the driver script of Figure 4.11, to solve numerically the three-node, two-member truss of Exercise 3.6. Verify that the output reproduces the solution given for that problem. Procedural recommendation: modify cells but keep a copy of the original Notebook handy in case things go wrong.

EXERCISE 4.6 [C:25] Expand the logic of `ModifiedMasterForcesForDBC` to permit specified nonzero displacements. Specify these in a second argument called `pval`, which contains a list of prescribed values paired with `pdof`.

```
xynode={{0,0},{10,0},{10,10}}; elenod={{1,2},{2,3},{3,1}};
unode={{0,0},{0,0},{2/5,-1/5}}; amp=5; p={};
For [t=0,t<=1,t=t+1/5,
  For [e=1,e<=Length[elenod],e++, {i,j}=elenod[[e]];
  xyi=xynode[[i]];ui=unode[[i]];xyj=xynode[[j]];uj=unode[[j]];
  p=AppendTo[p,Graphics[Line[{xyi+amp*t*ui,xyj+amp*t*uj}]]];
  ];
Show[p,Axes->False,AspectRatio->Automatic];
```

FIGURE E4.1. Mystery script for Exercise 4.7.

EXERCISE 4.7 [C:20] Explain what the program of Figure E4.1 does, and the logic behind what it does. (You may want to put it in a cell and execute it.) What modifications would be needed so it can be used for any plane stress?